[Honor 7X  Guides, News, & Discussion] mrmazak:
I like to try and Turn these monotonous repetitive things into tools.Mainly because one simple typo in the process can make it difficult to repair.

So I am in the process of making such a tool.

As a starting point I have made simple coding to do the recovery install part.

For now there are two batch programs , one for nougat , and one for oreo.

I am working toward some crafty line of code that can determine what build is on your devices first and according to the result choose which of the two batches to run.

In the meantime here are the two recovery switching batches. with recovery images included that have been copied from the xda forum.

@echo off
title	Lazy Recovery Replace Oreo
adb shell getprop ro.build.version.emui > %~dp0\version-info.txt
for /f %%i in ('FINDSTR "EmotionUI_" %~dp0\version-info.txt') do set emui=%%i
echo %emui%
set str=%emui:~10%
echo.%str%
pause
if %str% lss 5.3 (goto nougat
)else (
echo ok to continue)
adb reboot bootloader
echo Wait Here untill fastboot mode Loads On Phone
SET PATH=%PATH%;"%~dp0\files\oreo"
pause
fastboot oem get-build-number 2> %~dp0\build-info.txt
for /f "tokens=2" %%i in ('findstr "^(bootloader)" "%~dp0\build-info.txt"') do set Device=%%i
for /f "tokens=3" %%i in ('findstr "^(bootloader)" "%~dp0\build-info.txt"') do set Build=%%i
echo Your Current Device is = %Device% %Build%
pause
:MAIN
cls
echo	Choose what you need to work on.
echo(
echo %Device% %Build%
echo	][************************************][
echo	][ 1. complete_twrp_ramdisk ][
echo	][************************************][
echo	][ 2. Oreo Stock from beta ][
echo	][************************************][
echo	][ 3. twrp_p10_lite_0.3 Encryt works ][
echo	][************************************][
echo	][ 4. Oreo No-Check ][
echo	][************************************][
echo	][ 5. Local Image ][
echo	][************************************][
echo	][ 6. Cancel Exit and Reboot ][
echo	][************************************][
echo(
echo For performing Update simplest option is choose #1
set /p env=Type your option [1,2,3,4,5,6] then press ENTER: || set env="0"
if /I %env%==1 set recovery=complete_twrp_ramdisk.img && goto flash
if /I %env%==2 set recovery=RECOVERY_RAMDIS.img && goto flash
if /I %env%==3 set recovery=twrp_p10_lite_0.3.img && goto flash
if /I %env%==4 set recovery=RecoveryNoCheckOreoHi6250.img && goto flash
if /I %env%==5 call scripts\oreo\oreo_local_image_select.bat || goto end
if /I %env%==6 fastboot reboot && goto :eof
echo(
echo %env% is not a valid option. Please try again!
PING -n 3 127.0.0.1>nul
goto MAIN
:flash
echo THE FOLLOWING FILE HAS BEEN SELECTED
echo %recovery%
echo Continue IF it is correct, Else close window to cancel
pause
fastboot flash recovery_ramdisk %recovery%
:end
echo RECOVERY SHOULD NOW BE FLASHED
echo GET READY TO PULL USB PLUG OUT AND HOLD VOLUME UP
echo RIGHT AFTER YOU PRESS BUTTON TO CONTINUE
pause
fastboot reboot
goto :eof
exit
:nougat
echo You are On NOUGAT DO NOT USE THIS
pause
exit

@echo off
title	Lazy Recovery Replace Nougat
adb shell getprop ro.build.version.emui > %~dp0\version-info.txt
for /f %%i in ('FINDSTR "EmotionUI_" %~dp0\version-info.txt') do set emui=%%i
echo %emui%
set str=%emui:~10%
echo.%str%
pause
if %str% gtr 5.2 (goto oreo
)else (
echo ok to continue)
adb reboot bootloader
echo Wait Here untill fastboot mode Loads On Phone
SET PATH=%PATH%;"%~dp0\files\oreo"
pause
fastboot oem get-build-number 2> %~dp0\build-info.txt
for /f "tokens=2" %%i in ('findstr "^(bootloader)" "%~dp0\build-info.txt"') do set Device=%%i
for /f "tokens=3" %%i in ('findstr "^(bootloader)" "%~dp0\build-info.txt"') do set Build=%%i
echo Your Current Device is = %Device% %Build%
pause
:MAIN
cls
echo	Choose what you need to work on.
echo(
echo %Device% %Build%
echo	][************************************][
echo	][ 1. twrp-honor ][
echo	][************************************][
echo	][ 2. Stock Recovery ][
echo	][************************************][
echo	][ 3. Stock Recovery 2 ][
echo	][************************************][
echo	][ 4. BND-NO-CHECK ][
echo	][************************************][
echo	][ 5. Other file from your PC ][
echo	][************************************][
echo	][ 6. Cancel Exit and Reboot ][
echo	][************************************][
echo(
set /p env=Type your option [1,2,3,4,5,6] then press ENTER: || set env="0"
if /I %env%==1 set recovery=twrp-honor.img && goto flash
if /I %env%==2 set recovery=recovery.img && goto flash
if /I %env%==3 set recovery=recovery2.img && goto flash2
if /I %env%==4 set recovery=BND-RECOVERY-NoCheck.img && goto flash
if /I %env%==5 call scripts\nougat\nougat_recovery_file_flash.bat || goto end
if /I %env%==6 fastboot reboot && goto :eof
echo(
echo %env% is not a valid option. Please try again!
PING -n 3 127.0.0.1>nul
goto MAIN
:flash
echo THE FOLLOWING FILE HAS BEEN SELECTED
echo %recovery%
echo Continue IF it is correct, Else close window to cancel
pause
fastboot flash recovery %recovery%
goto end
:flash2
echo THE FOLLOWING FILE HAS BEEN SELECTED
echo %recovery%
echo Continue IF it is correct, Else close window to cancel
fastboot flash recovery2 %recovery%
:end
echo RECOVERY SHOULD NOW BE FLASHED
echo GET READY TO PULL USB PLUG OUT AND HOLD VOLUME UP
echo RIGHT AFTER YOU PRESS BUTTON TO CONTINUE
pause
fastboot reboot
goto :eof
exit
:oreo
echo You are On OREO DO NOT USE THIS
pause
exit

I have the launcher batch ready, it will detect the emui version with getprop. And call the appropriate switcher batch.

@echo off
if not defined in_subprocess (cmd /k set in_subprocess=y ^& %0 %*) & exit )
title	Lazy Recovery Auto Launcher
echo Waiting For device to be recognized by ADB
adb wait-for-device
adb shell getprop ro.build.version.emui > %~dp0\version-info.txt
for /f %%i in ('FINDSTR "EmotionUI_" %~dp0\version-info.txt') do set emui=%%i
echo %emui%
set str=%emui:~10%
echo.%str%
pause
if %str% equ 8 call scripts\oreo\Oreo_lazy_Recovery.bat
if %str% equ 8.1 call scripts\oreo\Oreo_lazy_Recovery.bat
if %str% equ 5 call scripts\nougat\Nougat_lazy_Recovery.bat
if %str% equ 5.1 call scripts\nougat\Nougat_lazy_Recovery.bat
pause
exit

Updated 5-21

Tool Download

.